Project: Vase Design

Douglas Meade, Ronda Sanders, and Xian Wu
Department of Mathematics

Instructions

The goal of the project is to design the most visually appealing vase that meets the following criteria:

- the vase will be molded using a symmetric mold, that is, the vase must be a solid of revolution;
- the function for the profile of the vase must be a piecewise-defined function with at least three "pieces", and at most one of the pieces can be a linear function; (Note that a constant function is a linear function);
- the thinnest part of the vase must be at least 1 inch in radius.
- You need to compute the following: the inside volume of the vase, the total materials used, and the surface area of the vase.

Your report should follow the guidelines set forth in the What is a Project Report/Grading Outline on our lab web. In particular, your report should include the following:

- a detailed description of your design.
- a (2-D) plot of the profile to be revolved and a (3-D) plot of the vase
- detailed numerical results as required.

Acknowledgment

The best design from each section and an over-all winner will be selected. We want to thank MapleSoft for providing some prizes for winning projects.

An Example in Maple:

For your convenience, this example is available in a Maple worksheet from our lab web.

Most maple commands used here were introduced in past labs. In particular, you should review last semester's Lab 6: (http://www.math.sc.edu/calclab/141L-F09/labs/RollerCoaster09f.pdf) Mathematical Models-Design a Roller Coaster and, of course, the last lab.

1. Basic Analyses: By studying our favorite vase at home, we decide that our vase should start with a base of 1.5 inches in radius and stands 9.5 inches tall. We would also like that its radius to be 1.1, 1.8, 1.6, and 2.5 inches at height of 2, 4, 6, and 9.5 inches, respectively. Put it in sideways and our vase hence extends from $x = 0$ to $x = 9.5$. The outside profile of our vase will then be described by a piecewise function that consists of two functions $f1(x)$ (a quadratic) and $f2(x)$ (a cubic) defined over $0 \le x < 4$ and $4 \le x \le 9.5$, respectively. Notice that while this example uses only two functions, you are asked to use at least three functions for your design.

2. Working with Maple

(a) You should always start a Maple session with `restart` command as it clears the internal memory so that Maple acts (almost) as if just started. In general, you need to start over from very beginning after changes are made.

```
> restart;
```

(b) Next, load needed packages using `with` command.

```
> with(plots):
> with(Student[Calculus1]):
```

(c) We first define $f1(x)$ and $f2(x)$ with constants a,b,c,d,e,f, and g to be solved later, and also find their derivatives.

```
> f1:=x->a*x^2+b*x+c;
> df1:=D(f1);
> f2:=x->d*x^3+e*x^2+f*x+g;
> df2:=D(f2);
```

(d) We need to set up seven equations to solve for seven constants. Five of them come from radius requirements at $x = 0, 2, 4, 6$,and 9.5. Smooth connection of two functions at $x = 4$ adds two more. (Note: the smoothness at connecting points is not necessary for vases.)

```
> eq1:=f1(0)=1.5;
> eq2:=f1(2)=1.1;
> eq3:=f1(4)=1.8;
> eq4:=f2(4)=f1(4);
> eq5:=df2(4)=df1(4);
> eq6:=f2(6)=1.6;
> eq7:=f2(9.5)=2.5;
```

(e) Now, let's solve for constants using `solve` command and assign solutions to values.

```
> values:=solve({eq1,eq2,eq3,eq4,eq5,eq6,eq7},{a,b,c,d,e,f,g});
```

(f) We can then plug those solved values into our functions once for all using `assign` command.

```
> assign(values);
```

(g) We are now ready to put two functions together as a piecewise function and plot it to see the outside profile.

```
> F:=x->piecewise(x<4,f1(x),x>=4,f2(x));
> F(x);
> plot(F(x),x=0..9.5,y=0..4,scaling=constrained);
```

(h) To get the piecewise function for the inside profile, we simply substrate 0.1 inches from the outside profile.

```
> G:=x->F(x)-0.1;
```

(i) Here is the vase. Rotate and/or right click the plot to try some visual effect options!

```
> VolumeOfRevolution(F(x),G(x),x=0..9.5,scaling=constrained, output=plot,
orientation=[0,180],title=``Example'');
```

(j) We still need to find numerical results as required. Please pay close attention as your TA will explain further details. (For example, you should use decimal limits for the surface area integral or Maple may not be able to finish the computation.)

```
> Capacity:=VolumeOfRevolution(G(x),0,x=0.1..9.5);
> Material:=VolumeOfRevolution(F(x),0,x=0..9.5)-Capacity;
> dF:=D(F);
> plot(G(x),x=0..9.5);
> fsolve(dF(x)=0,x=7..8);
> G(%);
>SurfaceArea:=int(2*Pi*F(x)*sqrt(1+(dF(x))^2),x=0..9.5);
```

3. Conclusion: Looks like the vase is too thin and its visual appearance can also be improved. Now, it is up to you to design a more interesting and better looking vase that satisfies all the criteria.